***The Feedback Effects of Controversial Police Use of Force
***Survey Implemented by Dynata June 21, 2019 - July 1, 2019

**********************************************
*Variable Recoding****************************
**********************************************

***Experimental Condition
tab FL_10_DO

gen group=.
replace group=1 if FL_10_DO=="Control"
replace group=2 if FL_10_DO=="NLText"
replace group=3 if FL_10_DO=="LText"
replace group=4 if FL_10_DO=="NLVideo"
replace group=5 if FL_10_DO=="LVideo"
replace group=6 if FL_10_DO=="NLTextVideo"
replace group=7 if FL_10_DO=="LTextVideo"
tab group
label define group 1 "Control" 2 "NonLethal Text" 3 "Lethal Text" 4 "NonLethal Video" 5 "Lethal Video" 6 "NonLethal Text+Video" 7 "Lethal Text+Video"
label values group group
tab group

***Demographics

*Gender
tab Q66
gen gender=.
replace gender=0 if Q66==1
replace gender=1 if Q66==2
replace gender=2 if Q66==3
tab gender
label define gender 0 "Male" 1 "Female" 2 "Other"
label values gender gender
tab gender

*Education
tab Q68
gen edu=Q68
tab edu

*Income
tab Q70
gen income=Q70
tab income

*Ideology. 1=extremely liberal, 7=extremely conservative
tab Q65
gen ideo=Q65
tab ideo
gen ideo3=.
replace ideo3=0 if ideo==1
replace ideo3=0 if ideo==2
replace ideo3=0 if ideo==3
replace ideo3=1 if ideo==4
replace ideo3=2 if ideo==5
replace ideo3=2 if ideo==6
replace ideo3=2 if ideo==7
tab ideo3
label define ideo3 0 "Liberal" 1 "Moderate" 2 "Conservative"
label values ideo3 ideo3
tab ideo3

*Race and ethnicity
*Q44 Are you Spanish, Hispanic, or Latino or none of these?
*Q49 Choose one or more races that you consider yourself to be. Check all that apply.
tab Q44
tab Q49
gen race=.
replace race=0 if Q44==2 & Q49=="1"
replace race=1 if Q44==2 & Q49=="2"
replace race=2 if Q49=="1,2"
replace race=2 if Q49=="1,2,4"
replace race=2 if Q49=="1,3"
replace race=2 if Q49=="1,4"
replace race=2 if Q49=="1,4,5"
replace race=2 if Q49=="1,5"
replace race=2 if Q49=="1,6"
replace race=2 if Q49=="2,3"
replace race=2 if Q49=="2,3,4"
replace race=2 if Q49=="2,3,6"
replace race=2 if Q49=="2,4"
replace race=3 if Q49=="3"
replace race=3 if Q49=="4"
replace race=3 if Q49=="5"
replace race=3 if Q49=="6"
replace race=3 if Q44==1 & Q49=="1"
replace race=3 if Q44==1 & Q49=="2"
tab race
label define race 0 "Non-hispanic White" 1 "Non-hispanic Black" 2 "Multiracial" 3 "Other"
label values race race
tab race

gen race2=.
replace race2=0 if race==0
replace race2=1 if race==1
tab race2
label define race2 0 "Non-hispanic White" 1 "Non-hispanic Black"
label values race2 race2
tab race2


*Victim of a crime
tab Q71
gen victim=Q71-1
tab victim
label define victim 0 "No" 1 "Yes"
label values victim victim
tab victim

*Worried about victimization. 1=not worried at all, 7=extremely worried
tab Q72
gen worryvic=Q72
tab worryvic

*Q73 How safe do you feel in your neighborhood? 
tab Q73
gen feelsafe=Q73
tab feelsafe

**Police experience
*Q74_1 Have you or someone close to you ever been stopped by police without good reason? 
*Q74_2 Have the police ever used insulting language toward you or someone close to you? 
*Q74_3 Have the police ever used excessive force against you or someone close to you?
*Combined into a summary index and recoded so higher values mean more of these bad experiences
tab Q74_1 
tab Q74_2 
tab Q74_3 

gen exp1=.
replace exp1=0 if Q74_1==2
replace exp1=1 if Q74_1==1
gen exp2=.
replace exp2=0 if Q74_2==2
replace exp2=1 if Q74_2==1
gen exp3=.
replace exp3=0 if Q74_3==2
replace exp3=1 if Q74_3==1
gen experience=exp1+exp2+exp3
tab experience

**Racial Resentment
*To what extent do you agree or disagree with the following statements?
*Q76_1 Irish, Italian, Jewish, and many other minority groups overcame prejudice and worked their way up. Black people should do the same without any special favors. (1) 
*Q76_2 Generations of slavery and discrimination have created conditions that make it difficult for black people to work their way out of the lower class (2) 
*Q76_3 Over the past few years, black people have gotten less than they deserve (3) 
*Q76_4 It’s really only a matter of some people not trying hard enough; if black people would only try harder they could be just as well-off as whites (4) 

tab Q76_1
tab Q76_2
tab Q76_3
tab Q76_4

gen resent1=Q76_1
gen resent2=8-Q76_2
gen resent3=8-Q76_3
gen resent4=Q76_4

gen resent= (resent1+resent2+resent3+resent4)/4
tab resent


*Familiar with Castile
*Q82 How familiar are you with circumstances surrounding the shooting of Philando Castile?
tab Q82

tab Q82 if(group==1)
tabstat Q82 if(group==1), stats(mean sem n)

**Neighborhood quality
*How often do you see the following things in your neighborhood? 
*Q75_1 Garbage in the streets
*Q75_2 Graffiti on the walls 
*Q75_3 People drinking alcohol on the street 
*Q75_4 People buying or selling drugs on the street 

tab Q75_1
tab Q75_2
tab Q75_3
tab Q75_4

gen neighborhood=(Q75_1+Q75_2+Q75_3+Q75_4)/4




***************************************
****Main text OLS Models, no controls
***************************************


***************************************************************
****Emotional Responses. Corresponds to Table 2 in the paper.**
***************************************************************


*Q33 To what extent do you feel angry right now?  1=not at all angry, 5=extremely angry
*Main text
tab Q33
gen anger=Q33
tabstat anger, by(group) stats(mean sem n)
regress anger i.group, robust
*outreg2 using Emotions_OLS_NoControls.doc, replace ctitle(Anger) label 


*Q29 To what extent do you feel anxious right now?  1=not at all anxious, 5=extremely anxious
*Main text
tab Q29
gen anxious=Q29
tabstat anxious, by(group) stats(mean sem n)
regress anxious i.group, robust
*outreg2 using Emotions_OLS_NoControls.doc, append ctitle(Anxiety) label 

*Q63 To what extent do you feel upset right now?  1=not at all upset, 5=extremely upset
*Main text
tab Q63
gen upset=Q63
tabstat upset, by(group) stats(mean sem n)
regress upset i.group, robust
*outreg2 using Emotions_OLS_NoControls.doc, append ctitle(Upset) label 


********************************************************************
****Attitudes Towards Police. Corresponds to Table 3 in the paper.**
********************************************************************

*Q30 To what extent do you approve or disapprove of the way the police in the United States are doing their job? 1=strongly disapprove, 7=strongly approve
*Main text
tab Q30
gen approval=Q30
regress approval i.group, robust
*outreg2 using ApproveTrust_OLS_NoControls.doc, replace ctitle(Approval) label 

*Q31 How much trust do you have in police officers to treat people fairly? 1= no trust at all, 7 =complete trust
*Main text
tab Q31
gen trust = Q31
regress trust i.group, robust
*outreg2 using ApproveTrust_OLS_NoControls.doc, append ctitle(Trust) label 

*Q39_1 To what extent do you agree or disagree with the following statements? People’s basic rights are well-protected by the police. 1=Strongly disagree, 7=strongly agree
*Main text
tab Q39_1
regress Q39_1 i.group, robust
*outreg2 using ApproveTrust_OLS_NoControls.doc, append ctitle(Rights Protected) label 

*Q39_2 To what extent do you agree or disagree with the following statements? Police can be trusted to make decisions that are right for the people in their communities  1=Strongly disagree, 7=strongly agree
*Main text
tab Q39_2
regress Q39_2 i.group, robust
*outreg2 using ApproveTrust_OLS_NoControls.doc, append ctitle(Trust Decisions) label 



*************************************************************************
****Beliefs about Excessive Force. Corresponds to Table 4 in the paper.**
*************************************************************************

*Q32 How often do you think that police use excessive force when responding to a possible criminal situation?  1=never, 5=almost always
*Main text
tab Q32
regress Q32 i.group, robust
*outreg2 using ExcessiveForce_OLS_NoControls.doc, replace ctitle(Q1) label 

*Q330 How confident are you that the police in this country are adequately trained to avoid the use of excessive force? 1=not at all confident, 7=extremely confident
*Main text
tab Q330
regress Q330 i.group, robust
*outreg2 using ExcessiveForce_OLS_NoControls.doc, append ctitle(Q2) label 

*Q38 In situations where police use excessive force, how confident are you that the officers involved will receive the appropriate legal consequences? 1= not at all confident, 7=extremely confident
*Main text
tab Q38
regress Q38 i.group, robust
*outreg2 using ExcessiveForce_OLS_NoControls.doc, append ctitle(Q3) label 

*Q40 To what extent are you concerned that you or someone you know might be the target of excessive force by police? 1=not concerened at all, 7=extremely concerned
*Main text
tab Q40
regress Q40 i.group, robust
*outreg2 using ExcessiveForce_OLS_NoControls.doc, append ctitle(Q4) label 



********************************************************************
****Support for Body Cameras. Corresponds to Table 5 in the paper.**
********************************************************************

*Q37 To what extent do you support or oppose requiring police officers to wear body cameras? 1=strongly oppose, 7=strongly support
*Main text
tab Q37
regress Q37 i.group, robust
*outreg2 using BodyCameras_OLS_NoControls.doc, replace ctitle(BodyCam) label 




*******************************************************************************
*******************************************************************************


*********************************
*Supplemental Materials
*********************************

*********************************
*Manipulation check and Timing***
*********************************

*Q83 Earlier, we provided you with information about a recent interaction between a police officer and a driver. What happened in that interaction
tab Q83
tab Q83 if(group==2)
tab Q83 if(group==3)
tab Q83 if(group==4)
tab Q83 if(group==5)
tab Q83 if(group==6)
tab Q83 if(group==7)

**Timing, time on page to submit
*NL text
tabstat Q6_Page_Submit, stat(mean p50 n)
*L text
tabstat Q23_Page_Submit, stat(mean p50 n)
*NL video
tabstat Q9_Page_Submit, stat(mean p50 n)
*L video
tabstat Q26_Page_Submit, stat(mean p50 n)
*NL text + video
tabstat Q15_Page_Submit, stat(mean p50 n)
*L text + video
tabstat Q31_Page_Submit, stat(mean p50 n)

*********************************
*****DV Information**************
*********************************

tabstat anger, stats(mean sem n)
tabstat anxious, stats(mean sem n)
tabstat upset, stats(mean sem n)
tabstat anger if(group==1), stats(mean sem n)
tabstat anxious if(group==1), stats(mean sem n)
tabstat upset if(group==1), stats(mean sem n)

tabstat approval, stats(mean sem n)
tabstat trust, stats(mean sem n)
tabstat Q39_1, stats(mean sem n)
tabstat Q39_2, stats(mean sem n)
tabstat approval if(group==1), stats(mean sem n)
tabstat trust if(group==1), stats(mean sem n)
tabstat Q39_1 if(group==1), stats(mean sem n)
tabstat Q39_2 if(group==1), stats(mean sem n)

tabstat Q32, stats(mean sem n)
tabstat Q330, stats(mean sem n)
tabstat Q38, stats(mean sem n)
tabstat Q40, stats(mean sem n)
tabstat Q32 if(group==1), stats(mean sem n)
tabstat Q330 if(group==1), stats(mean sem n)
tabstat Q38 if(group==1), stats(mean sem n)
tabstat Q40 if(group==1), stats(mean sem n)

tabstat Q37, stats(mean sem n)
tabstat Q37 if(group==1), stats(mean sem n)

correlate anger anxious upset approval trust Q39_1 Q39_2 Q32 Q330 Q38 Q40 Q37

************************
****Orderd Logit Models
************************

**Emotional Reactions, OLogit
ologit anger i.group
*outreg2 using Emotions_OLogit_NoControls.doc, replace ctitle(Anger) label 

ologit anxious i.group
*outreg2 using Emotions_OLogit_NoControls.doc, append ctitle(Anxiety) label 

ologit upset i.group
*outreg2 using Emotions_OLogit_NoControls.doc, append ctitle(Upset) label 


**Attitudes Towards Police, OLogit
ologit approval i.group
*outreg2 using ApproveTrust_OLogit_NoControls.doc, replace ctitle(Approval) label 

ologit trust i.group
*outreg2 using ApproveTrust_OLogit_NoControls.doc, append ctitle(Trust) label 

ologit Q39_1 i.group
*outreg2 using ApproveTrust_OLogit_NoControls.doc, append ctitle(Rights Protected) label 

ologit Q39_2 i.group
*outreg2 using ApproveTrust_OLogit_NoControls.doc, append ctitle(Trust Decisions) label 


**Beliefs about Excessive Force, OLogit
ologit Q32 i.group
*outreg2 using ExcessiveForce_OLogit_NoControls.doc, replace ctitle(Q1) label 

ologit Q330 i.group
*outreg2 using ExcessiveForce_OLogit_NoControls.doc, append ctitle(Q2) label 

ologit Q38 i.group
*outreg2 using ExcessiveForce_OLogit_NoControls.doc, append ctitle(Q3) label 

ologit Q40 i.group
*outreg2 using ExcessiveForce_OLogit_NoControls.doc, append ctitle(Q4) label 


**Support for Body Cameras, OLogit
ologit Q37 i.group
*outreg2 using BodyCameras_OLogit_NoControls.doc, replace ctitle(BodyCam) label



*****************************
****OLS Models with Controls
*****************************

**Emotional Reactions, OLS Controls
regress anger i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using Emotions_OLS_Controls.doc, replace ctitle(Anger) label 

regress anxious i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using Emotions_OLS_Controls.doc, append ctitle(Anxiety) label 

regress upset i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using Emotions_OLS_Controls.doc, append ctitle(Upset) label 


**Attitudes Towards Police, OLS Controls
regress approval i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ApproveTrust_OLS_Controls.doc, replace ctitle(Approval) label 

regress trust i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ApproveTrust_OLS_Controls.doc, append ctitle(Trust) label 

regress Q39_1 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ApproveTrust_OLS_Controls.doc, append ctitle(Rights Protected) label 

regress Q39_2 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ApproveTrust_OLS_Controls.doc, append ctitle(Trust Decisions) label 


**Beliefs about Excessive Force, OLS Controls
regress Q32 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ExcessiveForce_OLS_Controls.doc, replace ctitle(Q1) label 

regress Q330 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ExcessiveForce_OLS_Controls.doc, append ctitle(Q2) label 

regress Q38 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ExcessiveForce_OLS_Controls.doc, append ctitle(Q3) label 

regress Q40 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using ExcessiveForce_OLS_Controls.doc, append ctitle(Q4) label 


**Support for Body Cameras, OLS Controls
regress Q37 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood, robust
*outreg2 using BodyCameras_OLS_Controls.doc, replace ctitle(BodyCam) label 


***************************************
****OLS Models by Race (with controls)
***************************************

**Emotional Responses by Race
regress anger i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Effects on Anger by Race") xscale(range(-0.25 1.25)) ylabel(-.5(.5)2.5) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export Anger_Race.pdf, replace

regress anxious i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Effects on Anxiety by Race") xscale(range(-0.25 1.25)) ylabel(-.5(.5)2.5) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export Anxiety_Race.pdf, replace

regress upset i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Effects on Upset by Race") xscale(range(-0.25 1.25)) ylabel(-.5(.5)2.5) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export Upset_Race.pdf, replace


**Attitudes Towards Police by Race
regress approval i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Approval) by Race") xscale(range(-0.25 1.25)) ylabel(-2(.5).25) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_Approval_Race.pdf, replace

regress trust i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Trust) by Race") xscale(range(-0.25 1.25)) ylabel(-1.5(.5).25) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_Trust_Race.pdf, replace

regress Q39_1 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Rights Protected) by Race") xscale(range(-0.25 1.25)) ylabel(-1.5(.5).25) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_Rights_Race.pdf, replace

regress Q39_2 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Trust Decisions) by Race") xscale(range(-0.25 1.25)) ylabel(-1.5(.5).25) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_TrustDecisions_Race.pdf, replace


**Beliefs about Excessive Force by Race
regress Q32 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (How Often) by Race") xscale(range(-0.25 1.25)) ylabel(-.75(.25)1) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_HowOften_Race.pdf, replace

regress Q330 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (Confidence in Training) by Race") xscale(range(-0.25 1.25)) ylabel(-1.5(.5).5) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_ConfTraining_Race.pdf, replace

regress Q38 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (Confidence in Consequences) by Race") xscale(range(-0.25 1.25)) ylabel(-1.5(.5).5) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_ConfConseq_Race.pdf, replace

regress Q40 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (Personal Concern) by Race") xscale(range(-0.25 1.25)) ylabel(-.25(.5)1.5) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_PersonalConcern_Race.pdf, replace


**Support for Body Cameras by Race
regress Q37 i.group i.gender edu income ideo i.race victim worryvic feelsafe experience resent neighborhood i.group##i.race, robust
margins, dydx(i.group) at(race=(0 1)) vsquish
marginsplot, scheme(s1mono)  title("Effect on Support for Body Cameras by Race") xscale(range(-0.25 1.25)) ylabel(-1(.5)1) yline(0,lpattern(dash))  xtitle("Race", margin(medium)) xlabel(0 "White" 1 "Black") ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export BodyCameras_Race.pdf, replace


***********************************************************************************************************
****OLS Models by Racial Resentment (with controls). Analysis restricted to non-hispanic white respondents
***********************************************************************************************************

**Emotional Responses by Racial Resentment
regress anger i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Effects on Anger by Racial Resentment") ylabel(0(.5)3.5) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export Anger_Resent.pdf, replace

regress anxious i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Effects on Anxiety by Racial Resentment") ylabel(0(.5)2.5) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export Anxiety_Resent.pdf, replace

regress upset i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Effects on Upset by Racial Resentment") ylabel(0(.5)3.5) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export Upset_Resent.pdf, replace

**Attitudes Towards Police by Racial Resentment
regress approval i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Approval) by Racial Resentment") ylabel(-2(.5).25) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_Approval_Resent.pdf, replace

regress trust i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Trust) by Racial Resentment") ylabel(-2(.5).25) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_Trust_Resent.pdf, replace

regress Q39_1 i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Rights Protected) by Racial Resentment") ylabel(-2(.5).25) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_Rights_Resent.pdf, replace

regress Q39_2 i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Police Attitudes (Trust Decisions) by Racial Resentment") ylabel(-2(.5).25) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export PA_TrustDecisions_Resent.pdf, replace


**Beliefs about Excessive Force by Racial Resentment
regress Q32 i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (How Often) by Racial Resentment") ylabel(-.5(.5)1) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_HowOften_Resent.pdf, replace

regress Q330 i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (Conf. in Training) by Racial Resentment") ylabel(-1.5(.5).5) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_ConfTraining_Resent.pdf, replace

regress Q38 i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (Conf. in Consequences) by Racial Resentment") ylabel(-1.5(.5).5) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_ConfConseq_Resent.pdf, replace

regress Q40 i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Excessive Force (Personal Concern) by Racial Resentment") ylabel(-1.5(.5).5) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export EF_PersonalConcern_Resent.pdf, replace


**Support for Body Cameras by Racial Resentment
regress Q37 i.group i.gender edu income ideo victim worryvic feelsafe experience c.resent neighborhood i.group#c.resent if(race==0), robust
margins, dydx(i.group) at(resent=(1 2 3 4 5 6 7)) vsquish
marginsplot, scheme(s1mono)  title("Effect on Support for Body Cameras by Racial Resentment") ylabel(-1(.5)1) yline(0,lpattern(dash))  xtitle("Racial Resentment (low to high)", margin(medium)) ytitle("Treatment Effects", margin(medium)) plot(,label("NonLethal Text" "Lethal Text" "NonLethal Video" "Lethal Video"  "NonLethal Text+Video" "Lethal Text+Video"))
*graph export BodyCameras_Resent.pdf, replace

